Solving BS PDE for call option 01-10-11

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[3]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Introduction

The original notebook was Transforming BS PDE into standard heat eqn 01-10-11.nb. I make some cosmetic changes.

I solve Black Scholes partial differential equation and derive the famous closed form expression for a European style call option. This work is a revised version of *BS PDE to Diffusion PDE 04-19-03.nb,* and roughly follows the treatment in the *The Mathematics of Financial Derivatives*, Wilmott, Howison, and Dewynne.

The Black Scholes PDE is

$$\frac{\partial f(S,t)}{\partial t} + r\,S\,\frac{\partial f(S,t)}{\partial S} + \frac{\sigma^2}{2} S^2 \frac{\partial^2 f(S,t)}{\partial S^2} = r\,f(S,t)$$

Rather than solve this equation directly, I transform it into the heat equation, and then use the Green's function solution of the heat equation to calculate the earlier time response to the payoff function of a call option struck at K and expiring at T

$$C(K, T) = \max(S_T - K,\ 0)$$

Calculation

Define the BS PDE

In[5]:=
$$w[1] = -r\,C[S, t] + D[C[S, t], t] + r\,S\,D[C[S, t], S] + \frac{\sigma^2}{2} S^2 D[C[S, t], \{S, 2\}]$$

Out[5]=
$$-r\,C[S, t] + C^{(0,1)}[S, t] + r\,S\,C^{(1,0)}[S, t] + \frac{1}{2} S^2 \sigma^2 C^{(2,0)}[S, t]$$

The transformation into the heat equation form will involve changes of both the independent variables S and t, and the dependent variable C(S, t).

Introduce a replacement rule for the general form of the new independent variables.

In[6]:=
$$w[2] = w[1] \;/.\; C \to ((C[x[\#1], \tau[\#2]])\;\&)$$

Out[6]=
$$-r\,C[x[S], \tau[t]] + \tau'[t]\,C^{(0,1)}[x[S], \tau[t]] + r\,S\,x'[S]\,C^{(1,0)}[x[S], \tau[t]] + \frac{1}{2} S^2 \sigma^2 \left(x''[S]\,C^{(1,0)}[x[S], \tau[t]] + x'[S]^2 C^{(2,0)}[x[S], \tau[t]]\right)$$

Introduce explicit expressions for the new independent variables

In[7]:=
$$\mathrm{def}[x] = x == \mathrm{Log}\left[\frac{S}{K}\right];$$
$$\mathrm{def}[\tau] = \tau == \frac{\sigma^2}{2}(T - t);$$

Using these expressions as rules

In[9]:=
$$w[3] = w[2] \;/.\; \left\{x \to \left(\left(\mathrm{Log}\left[\frac{\#}{K}\right]\right)\&\right),\ \tau \to \left(\left(\frac{\sigma^2}{2}(T - \#)\right)\&\right)\right\} \;//\; \mathrm{Expand}$$

Out[9]=
$$-r\,C\left[\mathrm{Log}\left[\frac{S}{K}\right], \frac{1}{2}(-t+T)\sigma^2\right] - \frac{1}{2}\sigma^2 C^{(0,1)}\left[\mathrm{Log}\left[\frac{S}{K}\right], \frac{1}{2}(-t+T)\sigma^2\right] + r\,C^{(1,0)}\left[\mathrm{Log}\left[\frac{S}{K}\right], \frac{1}{2}(-t+T)\sigma^2\right] - \frac{1}{2}\sigma^2 C^{(1,0)}\left[\mathrm{Log}\left[\frac{S}{K}\right], \frac{1}{2}(-t+T)\sigma^2\right] + \frac{1}{2}\sigma^2 C^{(2,0)}\left[\mathrm{Log}\left[\frac{S}{K}\right], \frac{1}{2}(-t+T)\sigma^2\right]$$

Note that this choice of time variable converts the PDE from a forward equation to a backward equation, as appropriate for options. The Black-Scholes PDE describes a diffusive process that starts at a future expiration time with a known terminal payoff. That payoff is then propagated backward in time with the solution of the PDE describing the present fair value of the payoff.

In[10]:=
```
w[4] =
 w[3] /. {Simplify[Solve[def[x], S][[1, 1]], x ∈ Reals] , Solve[def[τ], t][[1, 1]]} //
   ExpandAll // PowerExpand
```

Out[10]=
$$-r\,C[x,\tau]-\frac{1}{2}\sigma^2 C^{(0,1)}[x,\tau]+r\,C^{(1,0)}[x,\tau]-\frac{1}{2}\sigma^2 C^{(1,0)}[x,\tau]+\frac{1}{2}\sigma^2 C^{(2,0)}[x,\tau]$$

A convenience substitution results in a simplified PDE with a single parameter.

In[11]:=
```
def[k] = k == 2 r / σ^2
```

Out[11]=
$$k == \frac{2\,r}{\sigma^2}$$

In[12]:=
```
w[5] = w[4] /. Solve[def[k], r][[1, 1]] // Factor // (#[[-1]]) &
```

Out[12]=
$$k\,C[x,\tau]+C^{(0,1)}[x,\tau]+C^{(1,0)}[x,\tau]-k\,C^{(1,0)}[x,\tau]-C^{(2,0)}[x,\tau]$$

The next step is to reexpress the dependent variable according to

$$C(x,\ \tau)\ =\ e^{\alpha x+\beta\tau}\,\mathbb{C}(x,\ \tau)$$

where the two parameters α and β are to be specified later

In[13]:=
```
w[6] = w[5] /. C → ((Exp[α #1 + β #2] ℂ[#1, #2]) &) // Factor // (#[[-1]]) &
```

Out[13]=
$$-k\,\mathbb{C}[x,\tau]-\alpha\,\mathbb{C}[x,\tau]+k\,\alpha\,\mathbb{C}[x,\tau]+\alpha^2\,\mathbb{C}[x,\tau]-\beta\,\mathbb{C}[x,\tau]-\mathbb{C}^{(0,1)}[x,\tau]-\mathbb{C}^{(1,0)}[x,\tau]+k\,\mathbb{C}^{(1,0)}[x,\tau]+2\,\alpha\,\mathbb{C}^{(1,0)}[x,\tau]+\mathbb{C}^{(2,0)}[x,\tau]$$

The standard heat equation is homogeneous (doesn't have a term in $\mathbb{C}(x,\ \tau)$) and also purely diffusive (does not a have term in $\frac{\partial\mathbb{C}(x,\,\tau)}{\partial x}$). I consider the values of α and β that will cancel these two terms.

In[14]:=
```
w[7] = {Coefficient[w[6], ℂ[x, τ]] == 0, Coefficient[w[6], ℂ^(1,0)[x, τ]] == 0}
```

Out[14]=
$$\{-k-\alpha+k\,\alpha+\alpha^2-\beta == 0,\ -1+k+2\,\alpha == 0\}$$

Solve for α and β.

In[15]:=
```
w[8] = Solve[w[7], {α, β}][[1]] // Factor
```

Out[15]=
$$\left\{\alpha\to\frac{1-k}{2},\ \beta\to-\frac{1}{4}\,(1+k)^2\right\}$$

On substituting these values into the transformed PDE

In[16]:=
```
w[9] = -w[6] == 0 /. w[8] // Expand
```

Out[16]=
$$\mathbb{C}^{(0,1)}[x,\tau]-\mathbb{C}^{(2,0)}[x,\tau] == 0$$

which is the standard form for the heat equation.

2 Solving the heat equation and deriving the classic Black-Scholes formula

As detailed in the notebook *Green's function solution of Heat Equation 01-11-11.nb*, the Greens's function solution of

$$\frac{\partial G(x,\ \tau; x_0,\ \tau_0)}{\partial \tau} - \frac{\partial^2 G(x,\ \tau; x_0,\ \tau_0)}{\partial x^2} = \delta(x - x_0)\,\delta(\tau - \tau_0)$$

is

$$G(x,\ \tau; x_0,\ \tau_0) = \frac{1}{2\sqrt{\pi\,\tau}}\,e^{-\frac{(x-x_0)^2}{4\,\tau}}$$

and the response to an initial condition $\mathbb{C}(x_0)$ is given by

$$\mathbb{C}(x,\ \tau) = \int_{-\infty}^{\infty} G(x,\ \tau; x_0,\ \tau_0)\,\mathbb{C}(x_0)\,dx_0$$

For the call option problem at hand, I construct the appropriate initial condition $\mathbb{C}(x_0)$ and then carry out this integration.

The payoff for the European call option is

```
In[17]:= w2[1] = C[S, T] == Max[S - K, 0]

Out[17]= C[S, T] == Max[0, -K + S]
```

This must be expressed in terms of the new variables.

```
In[18]:= w2[2] = w2[1] /. C -> ((C[x[#1], τ[#2]]) &) /.
            {x → ((Log[#/K]) &), τ → ((σ^2/2 (T - #)) &)} // Expand

Out[18]= C[Log[S/K], 0] == Max[0, -K + S]
```

```
In[19]:= w2[3] =
          w2[2] /. {Simplify[Solve[def[x], S][[1, 1]], x ∈ Reals], Solve[def[τ], t][[1, 1]]} //
           PowerExpand

Out[19]= C[x, 0] == Max[0, -K + e^x K]
```

Proceeding with the dependent variable transformation

In[20]:=
```
w2[4] = w2[3] /. C → ((Exp[α #1 + β #2] ℂ[#1, #2]) &)
```
Out[20]=
$$e^{x\alpha}\,\mathbb{C}[x,0] == \mathrm{Max}[0, -K + e^{x} K]$$

In[21]:=
```
w2[5] = Solve[w2[4], ℂ[x, 0]][[1, 1]] /. Rule → Equal
```
Out[21]=
$$\mathbb{C}[x,0] == e^{-x\alpha}\,\mathrm{Max}[0, -K + e^{x} K]$$

With this initial condition, I evaluate the Green's function integral.

I use Int instead of the *Mathematica* Integrate to suppress the integration algorithms until some manipula tions can be performed.

In[22]:=
$$\texttt{w3[1] = Int}\left[\mathbb{C}[x0,0]\ \frac{1}{2\sqrt{\pi\tau}}\ \mathrm{Exp}\left[-\frac{(x-x0)^2}{4\tau}\right],\ \{x0,-\infty,\infty\}\right]\ \texttt{/.}\ (\texttt{w2[5] /. x → x0 /. Equal → Rule})$$

Out[22]=
$$\mathrm{Int}\left[\left(e^{-x0\,\alpha-\frac{(x-x0)^2}{4\tau}}\,\mathrm{Max}[0, -K+e^{x0}K]\right)\Big/\left(2\sqrt{\pi}\sqrt{\tau}\right),\ \{x0,-\infty,\infty\}\right]$$

Notice that the integrand is zero unless $e^{y} > 1$, or $y > 0$. I use some pattern matching rules to introduce these simplifications of the general integral

In[23]:=
```
w3[2] = w3[1] /. Max[0, a_] → a /. {x0, -∞, ∞} → {x0, 0, ∞} // ExpandAll
```
Out[23]=
$$\mathrm{Int}\left[-\frac{e^{-x0\,\alpha-\frac{x^2}{4\tau}+\frac{x\,x0}{2\tau}-\frac{x0^2}{4\tau}}K}{2\sqrt{\pi}\sqrt{\tau}}+\frac{e^{x0-x0\,\alpha-\frac{x^2}{4\tau}+\frac{x\,x0}{2\tau}-\frac{x0^2}{4\tau}}K}{2\sqrt{\pi}\sqrt{\tau}},\ \{x0,0,\infty\}\right]$$

At this point I invoke the *Mathematica* integration algorithms and simplify the result for τ is real and positive.

In[24]:=
```
w3[3] = w3[2] /. Int → Integrate //
   Simplify[#, Assumptions → {τ ∈ Reals, τ > 0}] & // PowerExpand
```
Out[24]=
$$\frac{1}{2}\,e^{\alpha\,(-x+(-2+\alpha)\,\tau)}\,K\left(e^{x+\tau}\left(1+\mathrm{Erf}\left[\frac{x-2(-1+\alpha)\,\tau}{2\sqrt{\tau}}\right]\right)+e^{2\alpha\tau}\left(-2+\mathrm{Erfc}\left[\frac{x-2\,\alpha\,\tau}{2\sqrt{\tau}}\right]\right)\right)$$

In the option theoretic literature the cumulative standard normal distribution $\mathcal{N}$ is preferred over Erf. Note that //. has to be used in order that the Erfc is transformed into Erf and then Erf is transformed into $\mathcal{N}$

In[25]:=
$$\texttt{w3[4] = w3[3] //. }\{\mathrm{Erfc}[x\_] \to 1 - \mathrm{Erf}[x],\ \mathrm{Erf}[x\_] \to 2\,\mathcal{N}[\sqrt{2}\,x] - 1\}\texttt{ // ExpandAll}$$

Out[25]=
$$-e^{-x\,\alpha+\alpha^2\,\tau}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\sqrt{\tau}}-\sqrt{2}\,\alpha\sqrt{\tau}\right]+e^{x-x\,\alpha+\tau-2\,\alpha\,\tau+\alpha^2\,\tau}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\sqrt{\tau}}+\sqrt{2}\sqrt{\tau}-\sqrt{2}\,\alpha\sqrt{\tau}\right]$$

where

$$\mathcal{N}(x) = \int_{-\infty}^{x} \mathrm{dt}\, n(t) \qquad\qquad n(t) = \frac{1}{\sqrt{2\,\pi}}\, e^{-\frac{t^2}{2}}$$

So I arrive at an explicit expression for the call option in terms of known functions

In[26]:=
```
w3[5] = ℂ[x, τ] == w3[4]
```

Out[26]=

$$\mathbb{C}[x, \tau] == -e^{-x\,\alpha+\alpha^2\,\tau}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\,\sqrt{\tau}} - \sqrt{2}\,\alpha\,\sqrt{\tau}\right] + e^{x-x\,\alpha+\tau-2\,\alpha\,\tau+\alpha^2\,\tau}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\,\sqrt{\tau}} + \sqrt{2}\,\sqrt{\tau} - \sqrt{2}\,\alpha\,\sqrt{\tau}\right]$$

This result needs to be expressed in terms of the original variables.

In[27]:=
```
w3[6] = (# Exp[α x + β τ]) & /@ w3[5] // Expand
```

Out[27]=

$$e^{x\,\alpha+\beta\,\tau}\,\mathbb{C}[x, \tau] == -e^{\alpha^2\,\tau+\beta\,\tau}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\,\sqrt{\tau}} - \sqrt{2}\,\alpha\,\sqrt{\tau}\right] + e^{x+\tau-2\,\alpha\,\tau+\alpha^2\,\tau+\beta\,\tau}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\,\sqrt{\tau}} + \sqrt{2}\,\sqrt{\tau} - \sqrt{2}\,\alpha\,\sqrt{\tau}\right]$$

Introduce the explicit values for α and β

In[28]:=
```
w3[7] = w3[6] /. e^(x α+β τ) ℂ[x, τ] → C[x, τ] /. w[8]  // ExpandAll
```

Out[28]=

$$C[x, \tau] == -e^{-k\,\tau}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\,\sqrt{\tau}} - \frac{\sqrt{\tau}}{\sqrt{2}} + \frac{k\,\sqrt{\tau}}{\sqrt{2}}\right] + e^{x}\,K\,\mathcal{N}\left[\frac{x}{\sqrt{2}\,\sqrt{\tau}} - \frac{\sqrt{\tau}}{\sqrt{2}} + \sqrt{2}\,\sqrt{\tau} + \frac{k\,\sqrt{\tau}}{\sqrt{2}}\right]$$

Transform back to the original independent variables

In[29]:=
```
w3[8] =
 w3[7] /. Solve[def[k], k]〚1, 1〛 /. Solve[def[x], x]〚1, 1〛 /. Solve[def[τ], τ]〚1, 1〛
```

Out[29]=

$$C\left[\mathrm{Log}\left[\frac{S}{K}\right], -\frac{1}{2}\,(t-T)\,\sigma^2\right] == -e^{r\,(t-T)}\,K\,\mathcal{N}\left[-\frac{1}{2}\,\sqrt{-(t-T)\,\sigma^2} + \frac{r\,\sqrt{-(t-T)\,\sigma^2}}{\sigma^2} + \frac{\mathrm{Log}\left[\frac{S}{K}\right]}{\sqrt{-(t-T)\,\sigma^2}}\right] + S\,\mathcal{N}\left[\frac{1}{2}\,\sqrt{-(t-T)\,\sigma^2} + \frac{r\,\sqrt{-(t-T)\,\sigma^2}}{\sigma^2} + \frac{\mathrm{Log}\left[\frac{S}{K}\right]}{\sqrt{-(t-T)\,\sigma^2}}\right]$$

Further manipulation is required to obtain the familiar form

In[30]:=
```
w3[9] = C[S, t] == Simplify[w3[8]〚2〛, {σ ∈ Reals, σ > 0}]
```

Out[30]=

$$C[S, t] == -e^{r\,(t-T)}\,K\,\mathcal{N}\left[\left(-(t-T)\,(2\,r-\sigma^2) + 2\,\mathrm{Log}\left[\frac{S}{K}\right]\right)\Big/\left(2\,\sqrt{-t+T}\,\sigma\right)\right] + S\,\mathcal{N}\left[\left(-(t-T)\,(2\,r+\sigma^2) + 2\,\mathrm{Log}\left[\frac{S}{K}\right]\right)\Big/\left(2\,\sqrt{-t+T}\,\sigma\right)\right]$$

I define the famous d_1 term

In[31]:=
```
def[d1] = d1 == w3[9][[2, 2]] /. a_.𝒩[b_] → b
```

Out[31]=
$$d1 == \left(-(t-T)\,(2\,r+\sigma^2)+2\,\mathrm{Log}\left[\frac{S}{K}\right]\right)\Big/\left(2\,\sqrt{-t+T}\,\sigma\right)$$

and solve for $\mathrm{Log}\left[\frac{S}{K}\right]$

In[32]:=
```
w3[10] = Solve[def[d1], Log[S/K]][[1, 1]]
```

Out[32]=
$$\mathrm{Log}\left[\frac{S}{K}\right]\to\frac{1}{2}\left(2\,r\,t-2\,r\,T+2\,d1\,\sqrt{-t+T}\,\sigma+t\,\sigma^2-T\,\sigma^2\right)$$

Substitute this result in the solution

In[33]:=
```
w3[11] = w3[9] /. w3[10]
```

Out[33]=
$$C[S,t] == -e^{r\,(t-T)}\,K\,\mathcal{N}\left[\frac{2\,r\,t-2\,r\,T+2\,d1\,\sqrt{-t+T}\,\sigma+t\,\sigma^2-T\,\sigma^2-(t-T)\,(2\,r-\sigma^2)}{2\,\sqrt{-t+T}\,\sigma}\right]+S\,\mathcal{N}\left[\frac{2\,r\,t-2\,r\,T+2\,d1\,\sqrt{-t+T}\,\sigma+t\,\sigma^2-T\,\sigma^2-(t-T)\,(2\,r+\sigma^2)}{2\,\sqrt{-t+T}\,\sigma}\right]$$

Simplify

In[34]:=
```
w3[12] = Simplify /@ w3[11]
```

Out[34]=
$$C[S,t] == S\,\mathcal{N}[d1]-e^{r\,(t-T)}\,K\,\mathcal{N}\left[d1-\sqrt{-t+T}\,\sigma\right]$$

and, at last, the classical Black Scholes formula for a European call option is obtained.

In another notebook this formula is derived by taking the expectation of the terminal payoff under a risk-neutral probability measure.